POWER・GLMPOWER
プロシジャによる症例数設計

東京理科大学　工学部
浜田知久馬

2006年5月31日 (水)
医薬・統計ワーキンググループ ユーザー会

症例数設計を行う際に必要な条件
2群の平均値の差のt検定

α	第一種の過誤 (有意水準) (通常は両側5%)
β	第二種の過誤(見逃す確率) (通常は20%)
SD	個体間のばらつきの大きさ
Δ (デルタ)	予想される平均値の差 (生物学的に検出したい差)

症例数設計の式

$$N = \frac{2\{z_{\alpha/2} + z_{\beta}\}^2 \, SD^2}{\Delta^2}$$

$z_{\bullet}$: 正規分布の上側 $\bullet$ 点

ex) $\alpha = 0.05,\ \beta = 0.20,\ SD = 20,\ \Delta = 10$

$$N = 2\{1.96 + 0.84\}^2 \times 20^2 / 10^2$$

$$= 62.8$$

プログラム

2標本の平均値の差の検定

```
proc power;
   twosamplemeans test=diff
   meandiff    =10   ←Δ
   stddev      =20   ←SD
   alpha       =0.05←α
   power       =0.8  ←1-β
   ntotal      = .;
```

POWERプロシジャの出力

Fixed Scenario Elements	
Distribution	Normal
Method	Exact
Alpha	0.05
Mean Difference	10
Standard Deviation	20
Nominal Power	0.8
Number of Sides	2

Computed N Total

Actual Power	N Total
0.801	128

What if situations



SD

8 10 12

22

20

18

pessimistic

10

20

optimistic

What if situations

```
proc power;
   twosamplemeans test=diff
   meandiff    =8,10,12
   stddev      =18,20,22
   alpha       =0.05
   power       =0.8
   ntotal      = .;
```

必要な症例数

Index	Computed Mean Diff	N Total Std Dev	Actual Power	N Total
1	8	18	0.803	162
2	8	20	0.804	200
3	8	22	0.801	240
4	10	18	0.801	104
5	10	20	0.801	128
6	10	22	0.800	154
7	12	18	0.808	74
8	12	20	0.804	90
9	12	22	0.802	108

POWERプロシジャの構文

```
PROC POWER;
   MULTREG;                重回帰分析
   ONECORR;                相関係数
   ONESAMPLEFREQ;          一標本比率の検定
   ONESAMPLEMEANS;         一標本平均値の検定
   ONEWAYANOVA;            一元配置分散分析
   PAIREDFREQ;             対応有りの比率の検定
   PAIREDMEANS;            対応有りの平均値の検定
   TWOSAMPLEFREQ;          二標本比率の検定
   TWOSAMPLEMEANS;         二標本平均値の検定
   TWOSAMPLESURVIVAL; 二標本生存時間の検定
```

最も単純な問題
―標本の割合の検定(がん第Ⅱ相試験)



H_0： $\pi = \pi_0$(閾値奏効率) 0.10

H_1： $\pi = \pi_1$(期待奏効率) 0.30

α： 第Ⅰ種の過誤(片側) 0.05

β： 第Ⅱ種の過誤 0.10

*実際にはSimonの2段階法が用いられる

N：症例数，X：奏効数

$$Z = \frac{X - N\pi_0}{\sqrt{N\pi_0(1-\pi_0)}}$$

$$N = \left(\frac{Z_\beta \sqrt{\pi_1(1-\pi_1)} + Z_\alpha \sqrt{\pi_0(1-\pi_0)}}{\pi_1 - \pi_0} \right)^2$$

$$= \left(\frac{1.28\sqrt{0.3(1-0.3)} + 1.64\sqrt{0.1(1-0.1)}}{0.30 - 0.10} \right)^2$$

$$= 29.1$$

POWERプロシジャのプログラム
正規近似

```
proc power ;
        onesamplefreq test=z
        method=normal
            sides             = 1
            alpha             = 0.05
            nullproportion    = 0.10
            proportion        = 0.30
            ntotal            = .
            power             = 0.90;
```

POWERプロシジャの出力
正規近似

Method	Normal approximation
Number of Sides	1
Null Proportion	0.1
Alpha	0.05
Binomial Proportion	0.3
Nominal Power	0.9

Computed N Total	
Actual Power	N Total
0.906	30

正規近似による検定

N ：症例数　　$= 30$, X ：奏効数　　,

$\pi_0 = 0.10$, $\pi_1 = 0.30$

H_0 : X　　$N(N\pi_0, N\pi_0(1-\pi_0)) = N(3, 2.7)$

H_1 : X　　$N(N\pi_1, N\pi_1(1-\pi_1)) = N(9, 6.3)$

棄却限界値　　$= N\pi_0 + Z_\alpha \sqrt{N\pi_0(1-\pi_0)}$

$= 30 \cdot 0.10 + 1.645 \cdot \sqrt{2.7} = 5.7$


p0
0.26
0.24
0.22
0.20
0.18
0.16
0.14
0.12
0.10
0.08
0.06
0.04
0.02
0.00
0
10
20
v
$N(\pi_1-\pi_0)=Z_\beta\sqrt{N\pi_1(1-\pi_1)}+Z_\alpha\sqrt{N\pi_0(1-\pi_0)}$
$Z_\alpha\sqrt{N\pi_0(1-\pi_0)}$
$Z_\beta\sqrt{N\pi_1(1-\pi_1)}$
$N(\pi_1-\pi_0)$

検出力曲線作成のプログラム
正規近似

```
proc power plotonly;
      onesamplefreq test=z method=normal
         sides            = 1
         alpha            = 0.05
         nullproportion = 0.10
         proportion       = 0.30
         ntotal           = 30
         power            =.;
        plot x=n min=2 max=50 step=1
         yopts=(ref=.9);
```


1.0
0.9
0.8
0.7
0.6
0.5
0.4
0.3
Power
0.9
0
10
20
30
40
50
Total Sample Size

二項分布(帰無仮説)


p0b
0.26
0.24
0.22
0.20
0.18
0.16
0.14
0.12
0.10
0.08
0.06
0.04
0.02
0.00
0
10
20
y

二項分布(対立仮説)




p0b
0.24
0.22
0.20
0.18
0.16
0.14
0.12
0.10
0.08
0.06
0.04
0.02
0.00
0
10
20
y

POWERプロシジャのプログラム
2項分布による正確な検定

```
proc power ;
        onesamplefreq method=exact
            sides             = 1
            alpha             = 0.05
            nullproportion    = 0.10
            proportion        = 0.30
            ntotal            = .
            power             = 0.90;
```

POWERプロシジャの出力
2項分布による正確な検定

```
ERROR: NTOTAL is not
 available as a result
 option for  METHOD=EXACT.
```

検出力曲線作成のプログラム
2項分布による正確な検定

```
proc power;
        onesamplefreq method=exact
           sides            = 1
           alpha            = 0.05
           nullproportion   = 0.10
           proportion       = 0.30
           ntotal           = 30
           power            =.;
           plot x=n min=2 max=50 step=1
           yopts=(ref=.9);
```


1.0
0.8
0.6
0.4
0.2
0
Power
0.9
0
10
20
30
40
50
Total Sample Size

大いなる疑問



第一種の過誤と検出力

Index	N Total	Lower Crit Val	Upper Crit Val	Actual Alpha	Power
1	30	.	7	0.0258	0.840
2	31	.	7	0.0306	0.865
3	32	.	7	0.0358	0.887
4	33	.	7	0.0417	0.906
5	34	.	7	0.0481	0.921
6	35	.	8	0.0200	0.867
7	36	.	8	0.0235	0.888
8	37	.	8	0.0274	0.905
9	38	.	8	0.0318	0.921
10	39	.	8	0.0366	0.934
11	40	.	8	0.0419	0.945


0.950
0.925
0.900
0.875
0.850
0.825
0.800
0.775
Power
0.9
26
28
30
32
34
36
38
40
Total Sample Size



2項検定に関する注意

1)離散分布ではαエラーは一定ではない.

2)棄却限界値がジャンプしたところで，検出力は下がり，その後漸増する.

3)離散分布の検定では，名義水準に近くなるようにNを設定する(locally optimal) 必要がある.

割合の2群間比較(優越性)

両側α　：　0.05

検出力　：　0.80

$\pi_1 = 0.45, \pi_2 = 0.60$

$\pi_2 - \pi_1 = 0.15$

$$\frac{\pi_2}{\pi_1} = 1.333, \frac{\pi_2/(1-\pi_2)}{\pi_1/(1-\pi_1)} = 1.8333$$

割合の2群間比較(優越性)

```
proc power;
        twosamplefreq test=pchi
           sides              = 2
           alpha              = 0.050
           grouppproportions = (.45,.60)
          npergroup = .
           power = 0.80;
```

割合の2群間比較(優越性)

Distribution	Asymptotic normal
Method	Normal approximation
Number of Sides	2
Alpha	0.05
Group 1 Proportion	0.45
Group 2 Proportion	0.6
Nominal Power	0.8
Null Proportion Difference	0

Computed N Per Group

Actual Power	N Per Group
0.800	173

割合の2群間比較(優越性)リスク差

```
proc power;
        twosamplefreq test=pchi
           sides            = 2
           alpha            = 0.050
           refproportion    = 0.45
           proportiondiff = 0.15
           npergroup = .
           power = 0.80;
```

割合の2群間比較(優越性)リスク比

```
proc power;
        twosamplefreq test=pchi
           sides             = 2
           alpha             = 0.050
           refproportion     = 0.45
           relativerisk      = 1.33333
           npergroup = .
           power = 0.80;
    run;
```

割合の2群間比較(優越性)オッズ比

```
proc power;
        twosamplefreq test=pchi
           sides              = 2
           alpha              = 0.050
           refproportion      = 0.45
           oddsratio          = 1.83333
           npergroup = .
           power = 0.80;
```

割合の２群間比較(非劣性)

```
proc power;
        twosamplefreq test=pchi
           sides            = u
           alpha            = 0.025
           refproportion    = 0.60
           proportiondiff = 0.05
           nullpdiff=-0.10
           npergroup = .
           power = 0.80;
```

割合の２群間比較(非劣性)

Distribution	Asymptotic normal
Method	Normal approximation
Number of Sides	U
Null Proportion Difference	-0.1
Alpha	0.025
Reference (Group 1) Proportion	0.6
Proportion Difference	0.05
Nominal Power	0.8

Computed N Per Group	
Actual Power	N Per Group
0.801	164

What if situations

```
proc power;
      twosamplefreq test=pchi
         sides            = u
         alpha            = 0.025
         refproportion    = 0.55,0.60,0.65
         proportiondiff   = 0,0.05
         nullpdiff=-0.10,-0.08
         npergroup =300
         power =.;
      plot min=100 max=500 yopts=(ref=.8,.9);
```

What if situations

Computed Power

Index	Null Proportion Diff	Ref Proportion	Proportion Diff	Power
1	-0.10	0.55	0.00	0.692
2	-0.10	0.55	0.05	0.961
3	-0.10	0.60	0.00	0.705
4	-0.10	0.60	0.05	0.967
5	-0.10	0.65	0.00	0.728
6	-0.10	0.65	0.05	0.975
7	-0.08	0.55	0.00	0.504
8	-0.08	0.55	0.05	0.897
9	-0.08	0.60	0.00	0.516
10	-0.08	0.60	0.05	0.908
11	-0.08	0.65	0.00	0.538
12	-0.08	0.65	0.05	0.925


Proportion Diff=0
1.0
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
Power
0.9
0.8
100
200
300
400
500
Sample Size Per Group
Null Proportion Diff
−0.1
−0.08
Ref Proportion
0.55
0.6
0.65


Proportion Diff=0.05
1.0
0.9
0.8
0.7
0.6
0.5
0.4
Power
0.9
0.8
100
200
300
400
500
Sample Size Per Group
Null Proportion Diff
−0.1
−0.08
Ref Proportion
0.55
0.6
0.65

GLMPOWER

一般線形モデル(General Linear Model)**についての症例数設計**

分散分析(要因の複数自由度の検定)

共分散分析

WEIGHT文によるアンバランスな症例数設計

CONTRAST**文による特定の対比の検定**

GLMPOWERの文法

PROC GLMPOWER < options > **;**

CLASS variables **;**

MODEL dependent-variables = effects **;**

WEIGHT variable **;**

CONTRAST 'label' effect values < ... effect values > < **/** options > **;**

POWER < options > **;**

PLOT < plot-options > < **/** graph-options > **;**


CITIZEN
2:24
NTT DoCoMo


ION SUPPLY DRINK
POCARI
SWEAT


COO-
H
C
OH
CH3

Hocking, R.R. (1985), The Analysis of Linear Models, Monterey, CA: Brooks/Cole Publishing Company.

- **長距離ランナーに対する電解質の補充が乳酸の産生を抑制するかを調べた研究**
- **4種類のドリンクを割り付けた後，10-mile走を行い，乳酸濃度を測定**

予想される実験結果と解析方針

Table 34.7: Mean Lactic Acid Buildup by Fluid

Water	EZD1	EZD2	LZ1	LZ2
35.6	33.7	30.2	29	25.9

Table 34.8: Planned Comparisons

	Contrast Coefficients				
Comparison	**Water**	**EZD1**	**EZD2**	**LZ1**	**LZ2**
Water versus electrolytes	4	-1	-1	-1	-1
EZD versus LZ	0	1	1	-1	-1
EZD1 versus EZD2	0	1	-1	0	0
LZ1 versus LZ2	0	0	0	1	-1

予測値の指定

```
data Fluids;
      input Fluid $ LacticAcid CellWgt;
      datalines;
            Water       35.6    1
            EZD1        33.7    1
            EZD2        30.2    1
            LZ1         29      1
            LZ2         25.9    1
      ;
```

GLMPOWERのプログラム

```
proc glmpower data=Fluids;
  class Fluid;
  model LacticAcid= Fluid;
  weight CellWgt;
  contrast “Water vs.oth" Fluid  -1 -1 -1 -1 4;
  contrast "EZD vs. LZ"   Fluid   1  1 -1 -1 0;
  contrast "EZD1 vs.EZD2“ Fluid   1 -1  0  0 0;
  contrast "LZ1 vs. LZ2"  Fluid   0  0  1 -1 0;
  power
    stddev=3.75 alpha=0.025 ntotal=. power=0.8;
```

GLMPOWERの出力

Computed N Total

Type	Source	Test DF	Error DF	Actual Power	N Total
Effect	Fluid	4	20	0.856	25
Contrast	Water vs. others	1	25	0.848	30
Contrast	EZD vs. LZ	1	35	0.848	40
Contrast	EZD1 vs. EZD2	1	110	0.813	115
Contrast	LZ1 vs. LZ2	1	140	0.810	145

GLMPOWERのプログラム

```
proc glmpower data=Fluids;
  class Fluid;
  model LacticAcid= Fluid;
  weight CellWgt;
  contrast "Water vs. others" Fluid  -1 -1 -1 -1 4;
  contrast "EZD vs. LZ"       Fluid   1  1 -1 -1 0;
  contrast "EZD1 vs. EZD2"    Fluid   1 -1  0  0 0;
  contrast "LZ1 vs. LZ2"      Fluid   0  0  1 -1 0;
  power
    stddev=3.75 alpha=0.025 ntotal=35 power=.;
    plot x=n min=20 max=100;
   run;
```

GLMPOWERの出力

Computed Power

Type	Source	Test DF	Power
Effect	Fluid	4	0.975
Contrast	Water vs. others	1	0.907
Contrast	EZD vs. LZ	1	0.788
Contrast	EZD1 vs. EZD2	1	0.285
Contrast	LZ1 vs. LZ2	1	0.224


Power
1.0
0.8
0.6
0.4
0.2
0
20
40
60
80
100
Total Sample Size
Source
Fluid
Water vs. others
EZD vs. LZ
EZD1 vs. EZD2
LZ1 vs. LZ2

GLMPOWER(2-way)のプログラム

```
data Fluids;
      input fluid $ dose LacticAcid CellWgt;
      datalines;
            EZD 1       33.7    1
            EZD 2       30.2    1
            LZ  1       29      1
            LZ  2       25.9    1
proc glmpower data=Fluids;
  class Fluid Dose; weight CellWgt;
  model LacticAcid= Fluid Dose Fluid*Dose;
    power stddev=3.75 alpha=0.025 ntotal=. power=0.80;
    plot x=power min=0.50 max=0.90;
   run;
```

GLMPOWER(2-way)の出力

		Computed N Total		
	Test	Error	Actual	N
Source	DF	DF	Power	Total
fluid	1	28	0.841	32
dose	1	48	0.802	52
fluid*dose	1	13368	0.800	13372


Total Sample Size
80
70
60
50
40
30
20
0.55
0.60
0.65
0.70
0.75
0.80
0.85
0.90
0.95
Power
Source
fluid
dose

Two-Way ANOVA with Covariate

Table 34.9: Mean Lactic Acid Buildup by Fluid and Altitude

	Fluid				
Altitude	**Water**	**EZD1**	**EZD2**	**LZ1**	**LZ2**
High	36.9	35.0	31.5	30	27.1
Low	34.3	32.4	28.9	27	24.7

予測値の指定

```
data Fluids2;
      input Altitude $ Fluid $ LacticAcid CellWgt;
      datalines;
            High        Water       36.9        2
            High        EZD1        35.0        1
            High        EZD2        31.5        1
            High        LZ1         30          1
            High        LZ2         27.1        1
            Low         Water       34.3        2
            Low         EZD1        32.4        1
            Low         EZD2        28.9        1
            Low         LZ1         27          1
            Low         LZ2         24.7        1
```

Two-Way ANOVA with Covariate

```
proc glmpower data=Fluids2;
      class Altitude Fluid;
      model LacticAcid = Altitude Fluid;
      weight CellWgt;
      contrast "Water vs. others" Fluid  -1 -1 -1 -1 4;
      contrast "EZD vs. LZ"       Fluid   1  1 -1 -1 0;
      contrast "EZD1 vs. EZD2"    Fluid   1 -1  0  0 0;
      contrast "LZ1 vs. LZ2"      Fluid   0  0  1 -1 0;
      power
         nfractional
         stddev      = 3.5
         ncovariates = 1
         corrxy      = 0.3 0
         alpha       = 0.025
         ntotal      = 100
         power       = .;
   plot min=50 max=200 yopts=(ref=.8);
```

共変量の調整

反応変数と相関係数が

ρ の共変量が存在

$$\sigma^2 \Rightarrow \sigma^2(1-\rho^2), \quad \sigma \Rightarrow \sigma\sqrt{(1-\rho^2)}$$

e.g.) $\rho = 0.3$

$$\sigma^2(1-0.3^2) = \sigma^2 \times 0.91$$

$$\sigma = \sigma\sqrt{0.91} = \sigma \times 0.954$$

Two-Way ANOVA with Covariate

Index	Type	Source	Corr XY	Adj Std Dev	Test DF	Power
1	Effect	Altitude	0.3	3.34	1	0.950
2	Effect	Altitude	0.0	3.50	1	0.929
3	Effect	Fluid	0.3	3.34	4	>.999
4	Effect	Fluid	0.0	3.50	4	>.999
5	Contrast	Water vs. others	0.3	3.34	1	>.999
6	Contrast	Water vs. others	0.0	3.50	1	>.999
7	Contrast	EZD vs. LZ	0.3	3.34	1	>.999
8	Contrast	EZD vs. LZ	0.0	3.50	1	>.999
9	Contrast	EZD1 vs. EZD2	0.3	3.34	1	0.771
10	Contrast	EZD1 vs. EZD2	0.0	3.50	1	0.728
11	Contrast	LZ1 vs. LZ2	0.3	3.34	1	0.491
12	Contrast	LZ1 vs. LZ2	0.0	3.50	1	0.450


Power
1.0
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
0.8
40
60
80
100
120
140
160
180
200
Total Sample Size
Source
Altitude
Fluid
Water vs. others
EZD vs. LZ
EZD1 vs. EZD2
LZ1 vs. LZ2
Corr XY
0.3
0

症例数設計のソフトウエアの機能比較

方法	SASV8	SASV9	Nquery5	UnifyP	StatXact	Sampsize
検定ベース						
一標本t検定	○	●	○	○		○
対応あるt検定	○	●	○	○		○
二標本t検定	○	●	○	○		○
Welch検定		●	○			
一標本2項検定	○	●	○	○	○	○
カイ2乗検定		●	○	○	○	
McNemars検定		●	○	○		○
CMH検定			○			
CA検定			○		○	
一元配置分散分析	○	●	○	○		
二元配置分散分析		●	○	○		
多元配置分散分析		●	○	○		
経時測定分散分析		●	○	○		
単回帰分析		●	○	○		○
重回帰分析		●	○	○		

症例数設計のソフトウエアの機能比較

方法	SASV8	SASV9	Nquery5	UnifyP	StatXact	Sampsize
ログランク検定		●	○			
一般化ｳｨﾙｺｸｿﾝ検定		●				
Tarone-Ware検定		●				
同等性検証						
一標本(計量値)	○	●	○	○		
二標本(計量値)	○	●	○	○		○
一標本(2値)	○	●	○	○		
二標本(2値)	○	●	○	○	○	○
信頼区間						
一標本(計量値)	○	●	○	○		
二標本(計量値)	○	●	○	○		
一標本(2値割合)	○	●	○	○		
二標本(2値割合)	○	●	○	○		
対応なし(オッズ比)			○	○		
対応あり(オッズ比)			○	○		

参考文献

1)Castelloe, J.M. (2000), “Sample size computations and power analysis with the SAS system.” Proceedings of the Twenty-Fifth Annual SAS Users Group International Conference, Paper 265-25.

2)Castelloe, J.M. and O’Brien, R.G. (2001), “Power and sample size determination for linear models.” Proceedings of the Twenty-Sixth Annual SAS Users Group International Conference, Paper 240-26.

3)浜田知久馬・藤井陽介(2003),“生存時間解析の症例数設計.” 日本ＳＡＳユーザー会2003論文集,73-100.

4)浜田知久馬(2005), POWERプロシジャによる症例数設計. SAS Forum ユーザー会 学術総会2005,127-152

5)Lakatos, E. (1988), "Sample sizes based on the log-rank statistic in complex clinical trials." Biometrics, 44, 229-241.

6)Bauer, D. and Lavery, R.(2004),”Proc Power in SAS 9.1."
Proceedings of the Twenty-ninth Annual SAS Users Group International Conference, Paper 195-29.

特別コース
SASによる症例数設計 (1日間)

- 6/26(月) 10:00〜17:00
- POWERとGLMプロシジャの概要
- 症例数設計の原理
- 基本的な症例数設計
- 臨床研究における症例数設計の実例

http://www.sas.com/offices/asiapacific/japan/training/course/samplesize.html